Package MC32 was designed for use by package MA32 which has itself been superseded; the use of this routine is not recommended.

1 SUMMARY

This subroutine **performs the arithmetic operations corresponding to one pivotal step** of Gaussian elimination.

ATTRIBUTES — Version: 1.0.0. **Types:** MC32A; MC32AD. **Calls:** None. **Language:** A Cray-2 assembler version is available and is more than twice as fast as the equivalent vectorized Fortran code. **Original date:** August 1986. **Origin:** I.S. Duff, Harwell and A. K. Dave, CRAY Research (UK) Ltd.

2 HOW TO USE THE PACKAGE

Each call to MC32 is equivalent to the execution of the Fortran code :

```
      DO 100 L=1,LFRONT
        IF (QQ(L).EQ.0.0) GO TO 100
        DO 50 K=1,KFRONT
          FA(K,L) = FA(K,L) + PVKOL(K)*QQ(L)
   50   CONTINUE
  100 CONTINUE
```

2.1 Argument list

The single precision version

```
CALL MC32A(FA,NFRONT,KFRONT,LFRONT,PVKOL,INCPV,QQ,INCQQ)
```

The double precision version

```
CALL MC32AD(FA,NFRONT,KFRONT,LFRONT,PVKOL,INCPV,QQ,INCQQ)
```

FA is a REAL (DOUBLE PRECISION in the D version) array of two dimensions, with leading dimension NFRONT. The user must set the leading KFRONT by LFRONT subarray to the values of the matrix entries prior to the Gaussian elimination step to be executed by the call to MC32A/AD. On exit, FA contains the values of the matrix entries after the Gaussian elimination step. Only the leading KFRONT by LFRONT subarray is modified by MC32A/AD.

NFRONT is an INTEGER variable which must be set on entry to the leading dimension of array FA. It is not altered by MC32A/AD.

KFRONT is an INTEGER variable which must be set on entry to the number of entries in the pivot column (excluding the pivot). It is not altered by MC32A/AD.

LFRONT is an INTEGER variable which must be set on entry to the number of entries in the pivot row (excluding the pivot). It is not altered by MC32A/AD.

PVKOL is a REAL array of length INCPV $\times$ KFRONT which must be set to the negative of the subdiagonal entries in the pivot column. PVKOL is not altered by MC32A/AD.

INCPV is an INTEGER variable. The entries of the pivot column are in positions 1 + INCPV $\times$ (i-1) , i=1, KFRONT of array PVKOL. It is not altered by MC32A/AD.

QQ is a REAL array of length INCQQ×LFRONT which must be set to the the superdiagonal entries in the pivot row divided by the pivot. QQ is not altered by MC32A/AD.

INCQQ is an INTEGER variable. The entries of the pivot row are in positions 1 + INCQQ × (i-1) , i=1, LFRONT of array QQ. It is not altered by MC32A/AD.

2.3 Common

None.

2.4 Errors and diagnostic messages

None.

3 GENERAL INFORMATION

Workspace: There are no work arrays used by MC32A/AD.

Use of common: The subroutine does not use any common areas or auxiliary subroutines.

Other routines called directly: None

Input/output: None

Portability: There is a CRAY-2 Assembler Language available.

Restrictions: None.

4 METHOD

In the CRAY version, CRAY Assembler Language (CAL) is used to ensure that the vector PVKOL is kept in the vector registers throughout the sequence of inner loops. Thus there is only one memory fetch/store for each arithmetic operation and a maximum asymptotic rate of nearly 185 Megaflops is possible.

5 EXAMPLE OF USE

The following example shows the use of the subroutine MC32A/AD to factorise a full matrix using Gaussian elimination without pivoting.

```
      PARAMETER ( NMAX = 20 , NMAXP1 = NMAX + 1 )
      REAL    A(NMAX,NMAXP1), PVKOL(NMAX), QQ(NMAX), RHS1, ZERO,
     +        TMAX, ERRMAX
      INTEGER I, II, K, N, NMK, NMKP1, POSITN
      EXTERNAL FA01AS
      REAL     FA01AS
      DATA ZERO/0.0E0/

C     GENERATE RANDOM MATRIX WITH RIGHT HAND SIDE AS COLUMN N+1.

      WRITE (6,99999)
      N = NMAX
      DO 20 I=1,N
        RHS1 = ZERO
        DO 10 J=1,N
          A(I,J) = FA01AS(-1)
          RHS1 = RHS1 + A(I,J)*FLOAT(J)
   10   CONTINUE
        A(I,N+1) = RHS1
```

```
   20 CONTINUE

C      EACH PASS THROUGH THIS LOOP PERFORMS
C      ONE STEP OF GAUSSIAN ELIMINATION.

       DO 70 K=1,N
         NMK = N - K
C        CREATE PVKOL
         DO 30 I=1,NMK
           PVKOL(I) = -A(K+I,K)
 30      CONTINUE
C        CREATE QQ
         NMKP1 = NMK + 1
         DO 50 I=1,NMKP1
           QQ(I) = A(K,K+I)/A(K,K)
C          STORE UPPER TRIANGLE READY FOR BACKSOLUTION.
           A(K,K+I) = QQ(I)
   50    CONTINUE
         IF (NMK.NE.0) THEN
C           PERFORM GAUSSIAN ELIMINATION
            CALL MC32A(A(K+1,K+1), NMAX, NMK, NMKP1, PVKOL, 1, QQ, 1)
         ENDIF
   70 CONTINUE

C      PERFORM BACK SUBSTITUTION

       DO 90 I=N-1,1,-1
         IP1 = I + 1
         DO 80 J=IP1,N
           A(I,N+1) = A(I,N+1) - A(I,J)*A(J,N+1)
80       CONTINUE
90     CONTINUE

C      PRINT MAXIMUM ERROR AND THE ASSOCIATED COMPONENT
       ERRMAX = ZERO
       POSITN = 0
       DO 110 I=1,N
         TMAX = AMAX1( ABS(A(I,N+1)-FLOAT(I)), ERRMAX)
         IF (TMAX .NE. ERRMAX) THEN
            POSITN = I
            ERRMAX = TMAX
         ENDIF
  110 CONTINUE
       WRITE(6,99998) ERRMAX , POSITN , A(POSITN,N+1)
99999 FORMAT (' OUTPUT FROM TEST OF MC32A ')
99998 FORMAT (' MAXIMUM ERROR  = ',E12.5,'  ATTAINED AT COMPONENT ',I2/
     +         ' VALUE OF COMPONENT  = ',E12.5 )
       STOP
       END
```

This produces the following output

```
OUTPUT FROM TEST OF MC32A
MAXIMUM ERROR  =  0.65995E-10  ATTAINED AT COMPONENT  1
VALUE OF COMPONENT  =  0.10000E+01
```